

## I – Automatisation plan d'amortissement linéaire

	A	B	C	D	E	F	G	H
1						Ne marche que si la date inventaire est le 31/12		
2			Coût d'acquisition :	10 000,00 €		Durée utilisation en 2014 :	345 jrs	
3			Date d'acquisition :	01/01/2014		Nb lignes :	5	
4			Date de mise en service :	15/01/2014		Taux d'amortissement :	25,00%	
5			Valeur résiduelle :	1 000,00 €		Base amortissable :	9 000,00 €	
6			Durée d'utilisation :	4 ans		Première annuité :	2 156,25 €	
7						Annuité entière :	2 250,00 €	
8								
9								
10								
11								
12								
13								
14								

  

Années	Amortissements	Amortissements Cumulés	VNC
2014	2 156,25	2 156,25	7 843,75
2015	2 250,00	4 406,25	5 593,75
2016	2 250,00	6 656,25	3 343,75
2017	2 250,00	8 906,25	1 093,75
2018	93,75	9 000,00	1 000,00

### Le code

```
'Automatise le plan d'amortissement linéaire
'AUTEUR : Erapa Jean Marie
'Ne marche que si la date inventaire est le 31/12

Sub Plan_Amortissement_Linéaire()
'Déclaration des variables

Dim CoutAcquisition As Single
Dim DateMiseEnService As Date
Dim ValeurRésiduelle As Single
Dim Durée As Integer
Dim i As Integer           ' i =Compteur pour les lignes du tableau

Dim DuréeUtilisationPremièreAnnée As Integer
Dim NbLignes As Integer   'Nombre de lignes du tableau
Dim TauxAmortissement As Single
Dim BaseAmortissable As Single
Dim PremièreAnnuité As Single      'Amortissement pour la 1ere année
Dim AnnuitéEntière As Single
Dim NbJoursPremièreAnnée As Integer      'Prorata temporis Nb jours utilisation 1ère année
Dim NbJoursDernièreAnnée As Integer      'Nb jours utilisation dernière année

Dim AmortissementsCumulés As Single

'Recupère les données
CoutAcquisition = Worksheets("Amt_Linéaire").Range("D2")
DateMiseEnService = Worksheets("Amt_Linéaire").Range("D4")
ValeurRésiduelle = Worksheets("Amt_Linéaire").Range("D5")
Durée = Worksheets("Amt_Linéaire").Range("D6")

'Contrôle si toutes les données ont été saisies pour lancer le calcul
'S il manque une seule donnée, on affiche un message et on quitte le programme
If CoutAcquisition = 0 Then
    MsgBox "Il manque le coût d'acquisition "
    Exit Sub
End If
If DateMiseEnService = 0 Then
    MsgBox "Il manque la date de mise en service "
    Exit Sub
End If
If Durée = 0 Then
    MsgBox "Il manque la durée "
    Exit Sub
End If
```

```

'Calcule les données de l'en tête
Worksheets("Amt_Linéaire").Range("G2") = "Durée utilisation en " & Year(DateMiseEnService) & " : "

'Calcule la durée utilisation première année
If Day(DateMiseEnService) <= 2 Then
    'Si le jour de la date de mise en service est 1 ou 2,
    ' on prend le mois entier
    NbJoursPremièreAnnée = (13 - Month(DateMiseEnService)) * 30
Else
    NbJoursPremièreAnnée = 30 - Day(DateMiseEnService)
    NbJoursPremièreAnnée = NbJoursPremièreAnnée + (12 - Month(DateMiseEnService)) * 30
End If
Worksheets("Amt_Linéaire").Range("H2") = NbJoursPremièreAnnée

'Calcule le nombre de lignes du tableau
'En cas de prorata la première année, le tableau a une ligne de plus que la durée d'utilisation
If NbJoursPremièreAnnée < 360 Then
    NbLignes = Durée + 1
Else
    NbLignes = Durée
End If
Worksheets("Amt_Linéaire").Range("H3") = NbLignes

'Calcule le taux amortissement
Worksheets("Amt_Linéaire").Range("H4") = 1 / Durée

'Calcule la base amortissable
BaseAmortissable = CoutAcquisition - ValeurRésiduelle
Worksheets("Amt_Linéaire").Range("H5") = BaseAmortissable

'Calcule la première annuité
PremièreAnnuité = Round(BaseAmortissable * 1 / Durée * NbJoursPremièreAnnée / 360, 2)
Worksheets("Amt_Linéaire").Range("H6") = PremièreAnnuité
'Calcule l'annuité entière
AnnuitéEntière = Round(BaseAmortissable * 1 / Durée, 2)
Worksheets("Amt_Linéaire").Range("H7") = AnnuitéEntière

'=====
'Supprime les lignes du plan d'amortissement
ActiveSheet.Range("C10", ActiveSheet.Range("C10").End(xlDown).End(xlToRight)).Select
Selection.ClearContents 'Efface contenu
Selection.Borders.LineStyle = 0 'Efface encadrement
'=====
'Remplit les lignes du plan d'amortissement
'Initialisation des variables
i = 0
AmortissementsCumulés = 0

While i < NbLignes
    'Remplit la colonne Années
    'La première année est placée en C10
    Worksheets("Amt_Linéaire").Range("C" & i + 10) = Year(DateMiseEnService) + i

    'Remplit la colonne Amortissements
    'Le 1er amortissement est placée en D10
    If i = 0 Then 'Première année
        Worksheets("Amt_Linéaire").Range("D" & i + 10) = PremièreAnnuité
    Else
        If i = NbLignes - 1 Then 'Dernière année
            Worksheets("Amt_Linéaire").Range("D" & i + 10) = AnnuitéEntière - PremièreAnnuité
        Else
            Worksheets("Amt_Linéaire").Range("D" & i + 10) = AnnuitéEntière
        End If
    End If

    'Remplit la colonne Amortissements Cumulés : Colonne E
    AmortissementsCumulés = AmortissementsCumulés + Worksheets("Amt_Linéaire").Range("D" & i + 10)
    Worksheets("Amt_Linéaire").Range("E" & i + 10) = AmortissementsCumulés

    'Remplit la colonne VNC : Colonne F
    Worksheets("Amt_Linéaire").Range("F" & i + 10) = CoutAcquisition - AmortissementsCumulés

    i = i + 1 'Passe à la ligne suivante
Wend

'Encadre le tableau
'Selectionne la plage
ActiveSheet.Range("C10", ActiveSheet.Range("C9").End(xlDown).End(xlToRight)).Select
'Encadre
With Selection.Borders
    .ColorIndex = 1
    .LineStyle = xlContinuous
End With

End Sub

```

## Travail à faire

1. Expliquer ce que fait la ligne de code suivante : `CoutAcquisition = Worksheets("Amt_Linéaire").Range("D2")`

2. Expliquer les lignes suivantes

```
If Durée = 0 Then  
    MsgBox "Il manque la durée "  
    Exit Sub  
End If
```

3. Expliquer ce que fait la ligne de code suivante :

```
Worksheets("Amt_Linéaire").Range("G2") = "Durée utilisation en " & Year(DateMiseEnService) & " : "
```

4. Expliquer les lignes suivantes

'Calcule la durée utilisation première année

```
If Day(DateMiseEnService) <= 2 Then    'Si le jour de la date de mise en service est 1 ou 2,  
    'on prend le mois entier  
    NbJoursPremièreAnnée = (13 - Month(DateMiseEnService)) * 30  
Else  
    NbJoursPremièreAnnée = 30 - Day(DateMiseEnService)  
    NbJoursPremièreAnnée = NbJoursPremièreAnnée + (12 - Month(DateMiseEnService)) * 30  
End If  
  
Worksheets("Amt_Linéaire").Range("H2") = NbJoursPremièreAnnée
```

5. Expliquer les lignes suivantes

'Initialisation des variables

```
i = 0  
AmortissementsCumulés = 0  
  
While i < NbLignes  
    'Remplit la colonne Années  
    'La première année est placée en C10  
    Worksheets("Amt_Linéaire").Range("C" & i + 10) = Year(DateMiseEnService) + i  
  
    'Remplit la colonne Amortissements  
    'Le 1er amortissement est placée en D10  
    If i = 0 Then    'Première année  
        Worksheets("Amt_Linéaire").Range("D" & i + 10) = PremièreAnnuité  
    Else  
        If i = NbLignes - 1 Then 'Dernière année  
            Worksheets("Amt_Linéaire").Range("D" & i + 10) = AnnuitéEntière - PremièreAnnuité  
        Else  
            Worksheets("Amt_Linéaire").Range("D" & i + 10) = AnnuitéEntière  
        End If  
    End If  
  
    'Remplit la colonne Amortissements Cumulés : Colonne E  
    AmortissementsCumulés = AmortissementsCumulés + Worksheets("Amt_Linéaire").Range("D" & i + 10)  
    Worksheets("Amt_Linéaire").Range("E" & i + 10) = AmortissementsCumulés  
  
    'Remplit la colonne VNC : Colonne F  
    Worksheets("Amt_Linéaire").Range("F" & i + 10) = CoutAcquisition - AmortissementsCumulés  
  
    i = i + 1    'Passe à la ligne suivante  
Wend
```

## II – Programme pour générer les écritures comptables

1. Copier le document Projets\_Générer\_Ecritures qui se trouve dans le dossier Ressources\_BTS\_2\Processus\_7 dans votre dossier

Objectifs : Générer des écritures comptables à partir de la liste de données

	B	C	D	E	F
1	Client	Date	Total HT	Total TVA	Total TTC
2	DORVAL - DORVAL	02/01/2015	439,77	9,24	449,01
3	SANDANCE - SANDANCE	02/01/2015	117,42	2,58	120

Ecritures générées

Date	Jrnl	Compte	Pieces	Libelles	Débit	Crédit
02/01/2015	VT	411DIV	F 15010001	DORVAL - DORVAL	449,01	
02/01/2015	VT	707100	F 15010001	DORVAL - DORVAL		439,77
02/01/2015	VT	445710	F 15010001	DORVAL - DORVAL		9,24
02/01/2015	VT	411DIV	F 15010002	SANDANCE - SANDANCE	120	
02/01/2015	VT	707100	F 15010002	SANDANCE - SANDANCE		117,42
02/01/2015	VT	445710	F 15010002	SANDANCE - SANDANCE		2,58

Chaque ligne de données donne une écriture sur 3 lignes.

2. Cliquer sur le bouton Générer\_Ecritures

J	K	L	M	N	O	P
Date	Jrnl	Compte	Pieces	Libelles	Débit	Crédit
02/01/2015	VT	411DIV	F 15010001	DORVAL - DORVAL	449,01	
02/01/2015						
02/01/2015	VT	411DIV	F 15010002	SANDANCE - SANDANCE	120	
02/01/2015						

Le code est incomplet car il manque les lignes Crédit (Pour le compte de ventes et le compte de TVA collectée).

'Créé par ERAPA JEAN MARIE

'CREE LES ECRITURES COMPTABLES POUR IMPORTATION DANS UN LOGICIEL DE COMPTABILITE

Sub Generer\_Ecritures()

Dim i As Integer 'Compteur pour le tableau des écritures

Dim j As Integer 'Compteur pour le tableau des données initiales

Dim LastCol As Integer

Dim LastRow As Integer

i = 2 ' La 1ère ligne du tableau des données initiales démarre à 2

j = 2 ' La 1ère ligne du tableau des écritures initiales démarre à 2

'Sélectionne la plage écritures existante pour la supprimer

LastCol = ActiveSheet.Range("J2").End(xlToRight).Column

LastRow = ActiveSheet.Cells(65536, LastCol).End(xlUp).Row

ActiveSheet.Range("J2", ActiveSheet.Cells(LastRow + 2, LastCol)).Select

Selection.Clear 'Supprime les données existantes

'Sélectionne la 1ère ligne du tableau de données

Worksheets("Factures").Range("A" & j).Select 'J2 est la 1ère cellule qui contient une donnée

'Faire tant que la cellule n'est pas vide

Do While ActiveCell <> ""

'1ère ligne (Debit)

'Place la date

Worksheets("Factures").Range("J" & i) = Worksheets("Factures").Range("C" & j)

'Place le journal

Worksheets("Factures").Range("K" & i) = "VT"

'Place le code client

Worksheets("Factures").Range("L" & i) = "411DIV"

'Place le N° de pièce

Worksheets("Factures").Range("M" & i) = Worksheets("Factures").Range("A" & j)

'Place le libellé

Worksheets("Factures").Range("N" & i) = Worksheets("Factures").Range("B" & j)

'Place le montant débit

Worksheets("Factures").Range("O" & i) = Worksheets("Factures").Range("F" & j)

```

'Passe a la ligne suivante (Crédit 707100)
Worksheets("Factures").Range("J" & i + 1) = Worksheets("Factures").Range("C" & j)

'==PARTIE DU CODE A COMPLETER ; Il manque 5 lignes

'Passe a la ligne suivante (Crédit 445710)
'==PARTIE DU CODE A COMPLETER ; Il manque 6 lignes

i = i + 3
j = j + 1
Worksheets("Factures").Range("A" & j).Select

Loop

'Formate la colonne Date
Range("J2:J" & i).Select
Selection.NumberFormat = "dd/mm/yyyy"

'Sélectionne la plage écritures existante pour l'encadrer

LastCol = ActiveSheet.Range("J2").End(xlToRight).Column
LastRow = ActiveSheet.Cells(65536, LastCol).End(xlUp).Row
ActiveSheet.Range("J2", ActiveSheet.Cells(LastRow + 2, LastCol + 1)).Select
With Selection.Borders
    .ColorIndex = 1
    .LineStyle = xlContinuous
End With

Range("A1").Select

```

End Sub

3. Compléter les lignes de codes
4. Enregistrer
5. Exécuter pour tester